

Use Proc Format To Apply Value Specific Suffixes To Your Data

Lara E.H. Guttadauro, Kendle International Inc., Cincinnati, OH

ABSTRACT

If you need to apply value specific suffixes to your data, proc format in SAS® is a great option. You have the choice of using the picture option once or the value option twice.

MOTIVATION

A quick and easy method was needed to add value specific suffixes to p-values for output reporting across multiple programs. The following table is an example of the desired output that was the motivation for this procedure.

Table #

	TRT A	TRT B	p-value
CRITERIA 1	##	##	<.001 ***
CRITERIA 2	##	##	0.001 ***
CRITERIA 3	##	##	0.003 **
CRITIERA 4	##	##	0.023 *
CRITERIA 5	##	##	0.071 +
CRITERIA 6	##	##	0.125
CRITERIA 7	##	##	-

***, **, *, + Statistical significance at p=0.001, 0.01, 0.05, and 0.10 respectively.
- Indicates p-value greater than 0.20.

RAW DATA

The raw data values for the p-values in this example are displayed below.

CRITERIA	RAW_P
1	0.0008
2	0.001
3	0.003
4	0.023
5	0.071
6	0.125
7	0.405

PROC FORMAT

The use of proc format is an excellent choice for dealing with this situation, but it can be pretty intimidating to figure out which option is the best choice.

USE 'PICTURE' ONCE

The picture option of proc format is one way to apply suffixes (even though there is no suffix option to apply).

The suffix values are included as part of the picture format. The dash that appears for the values larger than 0.2 can be used directly in the picture format since it contains only characters. Since the picture for values less than 0.001 contains numbers, the noedit option must be applied in order for the number text to be displayed correctly (trying to display number text without the noedit option causes lots of headaches).

```
proc format;
  picture PEAS low -< 0.001 = '<.001 ***'(noedit)
           0.001 -< 0.01  = '9.999 ** '
           0.01  -< 0.05  = '9.999 *  '
           0.05  -< 0.1   = '9.999 +  '
           0.1   -< 0.2   = '9.999   '
           Other      = ' -          ';
run;
```

```

data PICTURE;
  set PVAL;
  PICTURE=put(RAW_P, PEAS.);
run;

```

USE 'VALUE' TWICE

The same results can be obtained by applying two formats using the VALUE option.

The default option sets the width of the values. This step is necessary to prevent extra spaces from appearing between the p-value and the suffix.

```

proc format;
  value PVALD(Default=5) 0-<0.2 = [Pvalue5.3]
                          Other = ' -';

  value SIGP(Default=4) Low   -< 0.001 = " ***"
                        0.001 -< 0.01  = " **"
                        0.01  -< 0.05  = " *"
                        0.05  -< 0.1   = " +"
                        Other          = " ";
run;

data DOUBLE;
  set PVAL;
  DOUBLE=put(RAW_P, PVALD.) || put(RAW_P, SIGP.);
run;

```

RESULTS & CONCLUSIONS

The same results are obtained using either the picture option one time or the value option two times. This makes it a toss up when deciding which option to use to apply value specific suffixes to data.

RESULT	RAW_P
<.001 ***	0.0008
0.001 **	0.0010
0.003 **	0.0030
0.023 *	0.0230
0.071 +	0.0710
0.125	0.1250
-	0.4050

ACKNOWLEDGMENTS

Thanks to Michael Bramley, Jeanne Phillipa and Richann Watson for their help in figuring out all the details.

CONTACT INFORMATION

Lara E.H. Guttadauro, MS
 Kendle International Inc.
 1200 Carew Tower
 441 Vine Street
 Cincinnati, OH 45202
 Phone: (513) 763-1363
 Email: guttadauro.lara@kendle.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.